```
> restart;
```

```
> ynp1:=a0*y(tn)+a1*y(tn-h)+a2*y(tn-2*h)+
      h*(b0*D(y)(tn)+b1*D(y)(tn-h)+
         b2*D(y)(tn-2*h))+E5*h^5*(D@@5)(y)(theta)/5!;
```

$$ynp1 := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) + \frac{E5\,h^5\,\mathrm{D}^{(5)}(y)(\theta)}{120} \tag{1}$$

```
> r:=y(tn+h)-ynp1;
```

$$r := y(tn+h) - a0\,y(tn) - a1\,y(tn-h) - a2\,y(tn-2\,h) - h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) - \frac{E5\,h^5\,\mathrm{D}^{(5)}(y)(\theta)}{120} \tag{2}$$

```
> eq[0]:=eval(subs(y=(x->1),r));
```

$$eq_0 := 1 - a0 - a1 - a2 \tag{3}$$

```
> for j from 1 to 5 do
     tmp[j]:=eval(subs(y=(x->x^j),r));
     eq[j]:=coeff(tmp[j],h^j);
     print(eq[j]);
  od:
```

$$1 + a1 + 2\,a2 - b0 - b1 - b2$$
$$1 - a1 - 4\,a2 + 2\,b1 + 4\,b2$$
$$1 + a1 + 8\,a2 - 3\,b1 - 12\,b2$$
$$1 - a1 - 16\,a2 + 4\,b1 + 32\,b2$$
$$1 + a1 + 32\,a2 - 5\,b1 - 80\,b2 - E5 \tag{4}$$

```
> S1:=solve({seq(eq[k],k=0..5)},{a0,a1,b0,b1,b2,E5});
```

$$S1 := \left\{E5 = \frac{40}{3} - \frac{4\,a2}{3},\ a0 = -8 - a2,\ a1 = 9,\ b0 = \frac{17}{3} + \frac{a2}{3},\ b1 = \frac{4\,a2}{3} + \frac{14}{3},\ b2 = -\frac{1}{3} + \frac{a2}{3}\right\} \tag{5}$$

```
> method:=subs(E5=0,ynp1);
```

$$method := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) \tag{6}$$

```
> m2:=subs(S1,method);
```

$$m2 := (-8 - a2)\,y(tn) + 9\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{17}{3} + \frac{a2}{3}\right)\mathrm{D}(y)(tn) + \left(\frac{4\,a2}{3} + \frac{14}{3}\right)\mathrm{D}(y)(tn-h) + \left(-\frac{1}{3} + \frac{a2}{3}\right)\mathrm{D}(y)(tn-2\,h)\right) \tag{7}$$

```
> m3:=eval(subs(D(y)=(x->f(x,y(x))),m2));
```

$$m3 := (-8 - a2)\,y(tn) + 9\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{17}{3} + \frac{a2}{3}\right) f(tn, y(tn)) + \left(\frac{4\,a2}{3} + \frac{14}{3}\right) f(tn-h, y(tn-h)) + \left(-\frac{1}{3} + \frac{a2}{3}\right) f(tn-2\,h, y(tn-2\,h))\right) \tag{8}$$

```
> f:=(xi,eta)->A*eta;
```

(9)

$$f := (\xi, \eta) \mapsto A \cdot \eta \tag{9}$$

```
> m4:=y(tn+h)=m3;
```

$$m4 := y(tn+h) = (-8-a2)\,y(tn) + 9\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{17}{3} + \frac{a2}{3}\right) A\,y(tn) + \left(\frac{4\,a2}{3} + \frac{14}{3}\right) A\,y(tn-h) + \left(-\frac{1}{3} + \frac{a2}{3}\right) A\,y(tn-2\,h)\right) \tag{10}$$

```
> ceq:=eval(subs(y=(s->rho^s),m4));
```

$$ceq := \rho^{tn+h} = (-8-a2)\,\rho^{tn} + 9\,\rho^{tn-h} + a2\,\rho^{tn-2\,h} + h\left(\left(\frac{17}{3} + \frac{a2}{3}\right) A\,\rho^{tn} + \left(\frac{4\,a2}{3} + \frac{14}{3}\right) A\,\rho^{tn-h} + \left(-\frac{1}{3} + \frac{a2}{3}\right) A\,\rho^{tn-2\,h}\right) \tag{11}$$

```
> ceq2:=subs({a2=1,tn=1,h=1},ceq);
```

$$ceq2 := \rho^2 = -9\,\rho + 9 + \frac{1}{\rho} + 6\,A\,\rho + 6\,A \tag{12}$$

```
> S2:=solve(ceq2,rho):
> Z1:=subs(A=a+I*b,abs(S2[1])):
> Z2:=subs(A=a+I*b,abs(S2[2])):
> Z3:=subs(A=a+I*b,abs(S2[3])):
> with(plots):
> contourplot(max(Z1,Z2,Z3),a=-3..3,b=-2..2,contours=[5,6,7,8,9,
  10],
     grid=[100,100],filled=true);
```




```
> # Note that max is >4 for all values of a and b.
  # Values of blue >10
> pred1:=subs(S1,subs(E5=0,ynp1));
```

$$pred1 := (-8-a2)\,y(tn) + 9\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{17}{3}\right. \right. \tag{13}$$

$$+\frac{a2}{3}\Big)\,\mathrm{D}(y)(tn)+\Big(\frac{4\,a2}{3}+\frac{14}{3}\Big)\,\mathrm{D}(y)(tn-h)+\Big(-\frac{1}{3}+\frac{a2}{3}\Big)\,\mathrm{D}(y)(tn-2\,h)\Big)$$